Arrakis Finance v2 Follow up II Audit Report

Dec 16, 2022



WATCHPUG

Table of Contents

Summary

This report has been prepared for Arrakis Finance v2 Follow up II Audit Report smart contract, to discover issues and vulnerabilities in the source code of their Smart Contract as well as any contract dependencies that were not part of an officially recognized library. A comprehensive examination has been performed, utilizing Static Analysis and Manual Review techniques.

The auditing process pays special attention to the following considerations:

- Testing the smart contracts against both common and uncommon attack vectors.
- Assessing the codebase to ensure compliance with current best practices and industry standards.
- Ensuring contract logic meets the specifications and intentions of the client.
- Cross referencing contract structure and implementation against similar smart contracts produced by industry leaders.
- Thorough line-by-line manual review of the entire codebase by industry experts.

Overview

Project Summary

Project Name	**Arrakis Finance v2 Follow up II Audit Report**
Codebase	**https://github.com/ArrakisFinance/vault-v2-core**
Commit	**026d9f346394b02b691be2b9259509abe386eab9**
Language	**Solidity**

Audit Summary

Delivery Date	**Dec 16, 2022**
Audit Methodology	**Static Analysis, Manual Review**
Total Isssues	**15**

[WP-M1] `_burnBuffer` mishandled the fee which could result in some users being unable to withdraw

Medium

Issue Description

L244-245 and L250-251 have not taken into account the fees belonging to the shareholders, `fee0` and `fee1` .

As a result, the additional amount to `leftOver` can be higher than `_burnBuffer` for small shareholders.

Thus, they may not be able to withdraw until `rebalance()` or until other users claim the fees first.

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2.sol#L135-L260

```
135  function burn(
136      BurnLiquidity[] calldata burns_,
137      uint256 burnAmount_,
138      address receiver_
139  ) external nonReentrant returns (uint256 amount0, uint256 amount1) {
     @@ 140,193 @@
194
195      Withdraw memory total;
196      {
197          for (uint256 i; i < burns_.length; i++) {
     @@ 198,224 @@
225          }
226
227          _applyFees(total.fee0, total.fee1);
228      }
229
230      if (amount0 > 0) {
231          token0.safeTransfer(receiver_, amount0);
232      }
233
```

```
234     if (amount1 > 0) {
235         token1.safeTransfer(receiver_, amount1);
236     }
237
238     // intentional underflow revert if managerBalance > contract's token balance
239     uint256 leftover0 = token0.balanceOf(address(this)) - managerBalance0;
240     uint256 leftover1 = token1.balanceOf(address(this)) - managerBalance1;
241
242     require(
243         (leftover0 <= underlying.leftOver0) ||
244             ((leftover0 - underlying.leftOver0) <=
245                 FullMath.mulDiv(total.burn0, _burnBuffer, hundredPercent)),
246         "L0"
247     );
248     require(
249         (leftover1 <= underlying.leftOver1) ||
250             ((leftover1 - underlying.leftOver1) <=
251                 FullMath.mulDiv(total.burn1, _burnBuffer, hundredPercent)),
252         "L1"
253     );
254
255     // For monitoring how much user burn LP token for getting their token back.
256     emit LPBurned(msg.sender, total.burn0, total.burn1);
257     emit LogUncollectedFees(underlying.fee0, underlying.fee1);
258     emit LogCollectedFees(total.fee0, total.fee1);
259     emit LogBurn(receiver_, burnAmount_, amount0, amount1);
260 }
```

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2.sol#L457-L479

```
457     function _withdraw(
458         IUniswapV3Pool pool_,
459         int24 lowerTick_,
460         int24 upperTick_,
461         uint128 liquidity_
462     ) internal returns (Withdraw memory withdraw) {
463         (withdraw.burn0, withdraw.burn1) = pool_.burn(
464             lowerTick_,
465             upperTick_,
466             liquidity_
```

```
467             );
468
469             (uint256 collect0, uint256 collect1) = pool_.collect(
470                 address(this),
471                 lowerTick_,
472                 upperTick_,
473                 type(uint128).max,
474                 type(uint128).max
475             );
476
477             withdraw.fee0 = collect0 - withdraw.burn0;
478             withdraw.fee1 = collect1 - withdraw.burn1;
479         }
```

PoC

Given:

- The total token0 holdings of the vault is `1000e18` ;
- The total unclaimed token0 fee is: `20e18` ;
- The token0 balance of the vault is: `1e18` , ie, `underlying.leftOver0 = 1e18` ;
- The total token0 holdings of Alice is `10e18` .
- `_burnBuffer` : 20%

1. Alice calls `burn()` to retrieve all her deposit. When `_withdraw()` is called, The vault receives `20e18` in fees while withdrawing `10e18` in liquidity, `total.burn0 = 10e18` ;
2. The current balance of the Vault becomes `20e18 + 10e18 + 1e18 == 31e18` ; After transfered `10e18` to Alice:
   `leftOver0 = 1e18(underlying.leftover0) + 10e18(burn0) + 20e18(fee0) - 10e18(Alice withdrawal) =`
   .

Unfortunately, this means that Alice cannot retrieve her money,

1. `leftover0 <= underlying.leftOver0` can not be satisfied.
2. `(leftover0 - underlying.leftOver0) == 20e18 <= FullMath.mulDiv(total.burn0, _burnBuffer, hundred`
   can not be satisfied, because `total.burn0` doesn't contain fee earned before, but leftover0 contains fee earned before.

Recommendation

Consider changing to:

```
fee0AfterManagerFee = (fee0_ * (hundredPercent - managerFeeBPS)) / hundredPercent;
fee1AfterManagerFee = (fee1_ * (hundredPercent - managerFeeBPS)) / hundredPercent;
require(
    (   fee0AfterManagerFee >= leftover0 ||
        leftover0 - fee0AfterManagerFee <= underlying.leftOver0) ||
        ((leftover0 - fee0AfterManagerFee - underlying.leftOver0) <=
            FullMath.mulDiv(total.burn0, _burnBuffer, hundredPercent)),
    "L0"
);
require(
    (   fee1AfterManagerFee >= leftover1 ||
        leftover1 - fee1AfterManagerFee <= underlying.leftOver1) ||
        ((leftover1 - fee1AfterManagerFee - underlying.leftOver1) <=
            FullMath.mulDiv(total.burn1, _burnBuffer, hundredPercent)),
    "L1"
);
```

Status

✓ Fixed

[WP-M2] `_rebalance()` Lack of slippage control for `burns`

Medium

Issue Description

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2.sol#L336-L455

```
336  function _rebalance(Rebalance calldata rebalanceParams_)
337      internal
338      nonReentrant
339  {
340      // Burns.
341      uint256 aggregator0 = 0;
342      uint256 aggregator1 = 0;
343      IUniswapV3Factory mFactory = factory;
344      address mToken0Addr = address(token0);
345      address mToken1Addr = address(token1);
346      for (uint256 i; i < rebalanceParams_.removes.length; i++) {
347          address poolAddr = mFactory.getPool(
348              mToken0Addr,
349              mToken1Addr,
350              rebalanceParams_.removes[i].range.feeTier
351          );
352          IUniswapV3Pool pool = IUniswapV3Pool(poolAddr);
353
354          Withdraw memory withdraw = _withdraw(
355              pool,
356              rebalanceParams_.removes[i].range.lowerTick,
357              rebalanceParams_.removes[i].range.upperTick,
358              rebalanceParams_.removes[i].liquidity
359          );
360
361          aggregator0 += withdraw.fee0;
362          aggregator1 += withdraw.fee1;
363      }
364
365      if (aggregator0 > 0 || aggregator1 > 0) {
366          _applyFees(aggregator0, aggregator1);
367
368          emit LogCollectedFees(aggregator0, aggregator1);
```

```
369         }
370
371         // Swap.
    @@ 372,452 @@
453
454         emit LogRebalance(rebalanceParams_);
455     }
```

The swap (the 2nd step) in `_rebalance` includes slippage control with `expectedMinReturn`.

However, the `Burns` are not controlled.

This means that a sudden market movement or an intentional frontrun price manipulation may result in a different output for the caller (the manager).

Specifically, a different `amountsOut` from the `burns`.

As a reference, the corresponding Uniswap v3 periphery `burn()` do have proper slippage control:

https://github.com/Uniswap/v3-periphery/blob/6cce88e63e176af1ddb6cc56e029110289622317/contracts/interfaces/INonfungiblePositionManager.sol#L139-L165

```
139 struct DecreaseLiquidityParams {
140     uint256 tokenId;
141     uint128 liquidity;
142     uint256 amount0Min;
143     uint256 amount1Min;
144     uint256 deadline;
145 }
146
147 /// @notice Decreases the amount of liquidity in a position and accounts it to the
    position
148 /// @param params tokenId The ID of the token for which liquidity is being
    decreased,
149 /// amount The amount by which liquidity will be decreased,
150 /// amount0Min The minimum amount of token0 that should be accounted for the
    burned liquidity,
151 /// amount1Min The minimum amount of token1 that should be accounted for the
    burned liquidity,
```

```
152  /// deadline The time by which the transaction must be included to effect the
     change
153  /// @return amount0 The amount of token0 accounted to the position's tokens owed
154  /// @return amount1 The amount of token1 accounted to the position's tokens owed
155  function decreaseLiquidity(DecreaseLiquidityParams calldata params)
156      external
157      payable
158      returns (uint256 amount0, uint256 amount1);
159
160  struct CollectParams {
161      uint256 tokenId;
162      address recipient;
163      uint128 amount0Max;
164      uint128 amount1Max;
165  }
```

Recommendation

Consider adding proper slippage control to the `burns`, similar to Uniswap v3's `NonfungiblePositionManager.sol`.

Status

✓ Fixed

[WP-I3] Incorrect/Misleading comment

Informational

Issue Description

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2.sol#L60-L126

```
60      function mint(uint256 mintAmount_, address receiver_)
61          external
62          nonReentrant
63          returns (uint256 amount0, uint256 amount1)
64      {
65          require(mintAmount_ > 0, "MA");
66          require(
67              restrictedMint == address(0) || msg.sender == restrictedMint,
68              "R"
69          );
70          address me = address(this);
71          uint256 ts = totalSupply();
72          bool isTotalSupplyGtZero = ts > 0;
73          (
74              uint256 current0,
75              uint256 current1,
76              uint256 fee0,
77              uint256 fee1
78          ) = isTotalSupplyGtZero
79                  ? UnderlyingHelper.totalUnderlyingWithFees(
80                      UnderlyingPayload({
81                          ranges: ranges,
82                          factory: factory,
83                          token0: address(token0),
84                          token1: address(token1),
85                          self: me
86                      })
87                  )
88                  : (init0, init1, 0, 0);
89          uint256 denominator = isTotalSupplyGtZero ? ts : 1 ether;
90
91          /// @dev current0 and current1 include fees and left over (but not manager
    balances)
```

```
92          amount0 = FullMath.mulDivRoundingUp(mintAmount_, current0, denominator);
93          amount1 = FullMath.mulDivRoundingUp(mintAmount_, current1, denominator);
94
95          // #region check amount0 is a multiple of current0.
96
97          if (!isTotalSupplyGtZero) {
98              uint256 amount0Mint = current0 != 0
99                  ? FullMath.mulDiv(amount0, denominator, current0)
100                 : type(uint256).max;
101             uint256 amount1Mint = current1 != 0
102                 ? FullMath.mulDiv(amount1, denominator, current1)
103                 : type(uint256).max;
104
105             require(
106                 (amount0Mint < amount1Mint ? amount0Mint : amount1Mint) ==
107                     mintAmount_,
108                 "A0&A1"
109             );
110         }
111
112         // #endregion check amount0 is a multiple of current0.
113
114         _mint(receiver_, mintAmount_);
115
116         // transfer amounts owed to contract
117         if (amount0 > 0) {
118             token0.safeTransferFrom(msg.sender, me, amount0);
119         }
120         if (amount1 > 0) {
121             token1.safeTransferFrom(msg.sender, me, amount1);
122         }
123
124         emit LogUncollectedFees(fee0, fee1);
125         emit LogMint(receiver_, mintAmount_, amount0, amount1);
126     }
```

The comment at L95/L112 is incorrect and misleading:

> #region check amount0 is a multiple of current0.

However, it does not required that `amount0` is a multiple of `current0`, in fact `amount0` is often not a multiple of `current0` and can still pass the check in L105.

Recommendation

Consider changing the comment to something like:

> #region check `amount0 * denominator - mintAmount * current0 < current0`

Status

✓ Fixed

[WP-M4] `VaultV2.burn()` may revert as the `BurnLiquidity[]` burns returned by `ArrakisV2Resolver.standardBurnParams()` can be slightly smaller than expected

Medium

Issue Description

If the total outAmounts from the burns (`BurnLiquidity[]`) returned by `ArrakisV2Resolver.standardBurnParams(amountToBurn_, vaultV2_)` is not enough, it may cause `vaultV2.burn()` to revert.

When all the `token0` and `token1` of the vault are in the liquidity of UniswapV3Pool (i.e., the vault contract itself has no token0 and token1 in its contract balance, and there is no pending fee in UniswapV3Pool), due to the accumulated precision loss of ArrakisV2Resolver at line 227, the total number of `token0` and `token1` taken out from UniswapV3Pool may not be enough, resulting in a revert at lines 231 and 235.

Furthermore, if ArrakisV2Resolver L227 rounds down to 0, `vaultV2.burn()` will revert at ArrakisV2 L198 as well.

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2Resolver.sol#L145-L238

```
145  function standardBurnParams(uint256 amountToBurn_, IArrakisV2 vaultV2_)
146      external
147      view
148      returns (BurnLiquidity[] memory burns)
149  {
     @@ 150,220 @@
221      burns = new BurnLiquidity[](len);
222      uint256 idx;
223      for (uint256 j; j < ranges.length; j++) {
224          if (liquidities[j] > 0) {
225              burns[idx] = BurnLiquidity({
226                  liquidity: SafeCast.toUint128(
227                      FullMath.mulDiv(
228                          liquidities[j],
```

```
                        amountToBurn_,
                        totalSupply
                    )
                ),
                range: ranges[j]
            });
            ++idx;
        }
    }
}
```

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2.sol#L135-L260

```
function burn(
    BurnLiquidity[] calldata burns_,
    uint256 burnAmount_,
    address receiver_
) external nonReentrant returns (uint256 amount0, uint256 amount1) {
@@ 140,192 @@
    _burn(msg.sender, burnAmount_);

    Withdraw memory total;
    {
        for (uint256 i; i < burns_.length; i++) {
            require(burns_[i].liquidity != 0, "LZ");
            {
                (bool exist, ) = Position.rangeExist(
                    ranges,
                    burns_[i].range
                );
                require(exist, "RRNE");
            }

            Withdraw memory withdraw = _withdraw(
                IUniswapV3Pool(
                    factory.getPool(
                        address(token0),
                        address(token1),
                        burns_[i].range.feeTier
```

```
213                     )
214                 ),
215                 burns_[i].range.lowerTick,
216                 burns_[i].range.upperTick,
217                 burns_[i].liquidity
218             );
219
220             total.fee0 += withdraw.fee0;
221             total.fee1 += withdraw.fee1;
222
223             total.burn0 += withdraw.burn0;
224             total.burn1 += withdraw.burn1;
225         }
226
227         _applyFees(total.fee0, total.fee1);
228     }
229
230     if (amount0 > 0) {
231         token0.safeTransfer(receiver_, amount0);
232     }
233
234     if (amount1 > 0) {
235         token1.safeTransfer(receiver_, amount1);
236     }
237

@@ 238,259 @@

260 }
```

Recommendation

Consider changing ArrakisV2Resolver L227 to `mulDivRoundingUp()` :

```
224 if (liquidities[j] > 0) {
225     burns[idx] = BurnLiquidity({
226         liquidity: SafeCast.toUint128(
227             FullMath.mulDivRoundingUp(
228                 liquidities[j],
229                 amountToBurn_,
230                 totalSupply
231             )
```

```
232             ),
233             range: ranges[j]
234         });
235         ++idx;
236     }
```

Status

✓ Fixed

[WP-I5] Consider adding `nonReentrant` modifier to `withdrawManagerBalance()`

Informational

Issue Description

https://github.com/ArrakisFinance/vault-v2-core/blob/d958ffd0e9ed7890b55d8ade4fdc26eae9640ab3/contracts/ArrakisV2.sol#L317-L333

```
317   function withdrawManagerBalance() external {
318       uint256 amount0 = managerBalance0;
319       uint256 amount1 = managerBalance1;
320
321       managerBalance0 = 0;
322       managerBalance1 = 0;
323
324       if (amount0 > 0) {
325           token0.safeTransfer(manager, amount0);
326       }
327
328       if (amount1 > 0) {
329           token1.safeTransfer(manager, amount1);
330       }
331
332       emit LogWithdrawManagerBalance(amount0, amount1);
333   }
```

The manager can reenter `burn()` if one of the tokens is a hookable token (ERC777) in `withdrawManagerBalance()`, and using the abnormal `pricePerShare` to withdraw more `token0` or token1 than expected.

Status

✓ Fixed

[WP-I6] Inconsistent `address(0)` check in `upgradeVaults()` and `upgradeVaultsAndCall()`

Informational

Issue Description

The `upgradeVaults()` function has been updated with an `implementation != address(0)` check, but the `upgradeVaultsAndCall()` function has not been updated.

By the way, consider using CAS to prevent the `arrakisV2Beacon.implementation()` from changing between the time the `upgradeVaults` transaction is sent and the time the transaction is minted.

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/abstract/ArrakisV2FactoryStorage.sol#L49-L55

```
49      function upgradeVaults(address[] memory vaults_) external onlyOwner {
50          address implementation = arrakisV2Beacon.implementation();
51          require(implementation != address(0), "implementation is address zero");
52          for (uint256 i = 0; i < vaults_.length; i++) {
53              ITransparentUpgradeableProxy(vaults_[i]).upgradeTo(implementation);
54          }
55      }
```

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/abstract/ArrakisV2FactoryStorage.sol#L62-L73

```
62      function upgradeVaultsAndCall(
63          address[] memory vaults_,
64          bytes[] calldata datas_
65      ) external onlyOwner {
66          require(vaults_.length == datas_.length, "mismatching array length");
67          for (uint256 i = 0; i < vaults_.length; i++) {
68              ITransparentUpgradeableProxy(vaults_[i]).upgradeToAndCall(
69                  arrakisV2Beacon.implementation(),
```

```
                datas_[i]
            );
        }
    }
```

Recommendation

Change to:

```
function upgradeVaults(address[] memory vaults_, address implementation_) external onlyOwner {
    address implementation = arrakisV2Beacon.implementation();
    require(implementation == implementation_, "implementation mismatch");
    for (uint256 i = 0; i < vaults_.length; i++) {
        ITransparentUpgradeableProxy(vaults_[i]).upgradeTo(implementation);
    }
}
```

```
function upgradeVaultsAndCall(
    address[] memory vaults_,
    bytes[] calldata datas_,
    address implementation_
) external onlyOwner {
    address implementation = arrakisV2Beacon.implementation();
    require(implementation == implementation_, "implementation mismatch");
    require(vaults_.length == datas_.length, "mismatching array length");
    for (uint256 i = 0; i < vaults_.length; i++) {
        ITransparentUpgradeableProxy(vaults_[i]).upgradeToAndCall(
            arrakisV2Beacon.implementation(),
            datas_[i]
        );
    }
}
```

Status

✓ Fixed

[WP-N7] Unused imports

Issue Description

ArrakisFinance/vault-v2-core

contracts/abstract/ArrakisV2FactoryStorage.sol

- ”Initializable” is unused

contracts/ArrakisV2.sol

- ”ERC20” is unused
- ”SafeCast” is unused

contracts/ArrakisV2Factory.sol

- ”ERC1967Upgrade” is unused

contracts/ArrakisV2Resolver.sol

- ”ERC20” is unused
- ”SwapPayload” is unused

PALM/PALMManager.sol

- ”VaultInfo” is unused

ArrakisFinance/v2-palm

contracts/abstracts/PALMManagerStorage.sol

- ”Ownable” is unused

contracts/PALMManager.sol

- ”VaultInfo” is unused

contracts/structs/SPALMTerms.sol

- ”BurnLiquidity” is unused

Status

✓ Fixed

[WP-D8] Improve the NatSpec comments for `totalUnderlyingWithFees()` about the returns

Issue Description

- `amount0` , `amount1` : the total amount of underlying tokens (`token0` and `token1`) in the ArrakisV2 vault corresponding to the share holders
- `fee0` , `fee1` : the amount of fees in the underlying UniswapV3Pool, where the `managerFeeBPS_` portion goes to the manager and the rest goes to the share holders.

Because, the return variable names are not specific and the scope of the `amount` and `fee` are not consistent, which can lead to misunderstandings or omissions of what is being returned.

Status

ⓘ Acknowledged

[WP-G9] `Position#rangeExist()` can be optimized

Gas

Issue Description

The MSTORE at L45 is only needed when `ok == true` :

https://github.com/ArrakisFinance/vault-v2-core/blob/d958ffd0e9ed7890b55d8ade4fdc26eae9640ab3/contracts/libraries/Position.sol#L35-L48

```
35      function rangeExist(Range[] memory currentRanges_, Range memory range_)
36          public
37          pure
38          returns (bool ok, uint256 index)
39      {
40          for (uint256 i; i < currentRanges_.length; i++) {
41              ok =
42                  range_.lowerTick == currentRanges_[i].lowerTick &&
43                  range_.upperTick == currentRanges_[i].upperTick &&
44                  range_.feeTier == currentRanges_[i].feeTier;
45              index = i;
46              if (ok) break;
47          }
48      }
```

Recommendation

```
35      function rangeExist(Range[] memory currentRanges_, Range memory range_)
36          public
37          pure
38          returns (bool ok, uint256 index)
39      {
40          for (uint256 i; i < currentRanges_.length; i++) {
41              ok =
42                  range_.lowerTick == currentRanges_[i].lowerTick &&
43                  range_.upperTick == currentRanges_[i].upperTick &&
44                  range_.feeTier == currentRanges_[i].feeTier;
45              if (ok) {
```

```
46                  index = i;
47                  break;
48              }
49          }
50      }
```

Status

✓ Fixed

[WP-G10] `safeApprove(0)` to the newly created vault is unnecessary

Gas

Issue Description

https://github.com/ArrakisFinance/v2-palm/blob/46546698096f0f4cd86381cd06be28d43a50a2ea/contracts/PALMTerms.sol#L56-L128

```
56  function openTerm(SetupPayload calldata setup_)
57      external
58      payable
59      override
60      collectLeftOver(setup_.token0, setup_.token1)
61      returns (address vault)
62  {
    @@ 63,107 @@
108     setup_.token0.safeTransferFrom(
109         msg.sender,
110         address(this),
111         setup_.amount0
112     );
113     setup_.token1.safeTransferFrom(
114         msg.sender,
115         address(this),
116         setup_.amount1
117     );
118
119     setup_.token0.safeApprove(vault, 0);
120     setup_.token1.safeApprove(vault, 0);
121
122     setup_.token0.safeApprove(vault, setup_.amount0);
123     setup_.token1.safeApprove(vault, setup_.amount1);
124
125     vaultV2.mint(mintAmount, address(this));
126
127     emit SetupVault(setup_.owner, vault);
128 }
```

The **vault** is a newly created address, therefore the existing allowance must be **zero**.

The `safeApprove` at L119-120 is redundant.

Recommendation

Remove L119-120.

Status

✓ **Fixed**

[WP-G11] `PALMManagerStorage._withdrawVaultBalance()` can be optimized

Gas

Issue Description

https://github.com/ArrakisFinance/v2-palm//blob/46546698096f0f4cd86381cd06be28d43a50a2ea/contracts/abstracts/PALMManagerStorage.sol#L480-L493

```
480     function _withdrawVaultBalance(
481         address vault_,
482         uint256 amount_,
483         address payable to_
484     ) internal {
485         require(
486             vaults[vault_].balance >= amount_,
487             "PALMManager: amount exceeds available balance"
488         );
489         vaults[vault_].balance -= amount_;
490         to_.sendValue(amount_);
491 
492         emit WithdrawVaultBalance(vault_, amount_, to_, vaults[vault_].balance);
493     }
```

Recommendation

Consider changing to:

```
480     function _withdrawVaultBalance(
481         address vault_,
482         uint256 amount_,
483         address payable to_
484     ) internal {
485         uint256 oldBalance = vaults[vault_].balance;
486         require(
487             oldBalance >= amount_,
488             "PALMManager: amount exceeds available balance"
```

```
489         );
490         uint256 newBalance = oldBalance - amount_;
491         vaults[vault_].balance = newBalance;
492         to_.sendValue(amount_);
493
494         emit WithdrawVaultBalance(vault_, amount_, to_, newBalance);
495     }
```

Status

✓ Fixed

[WP-G12] Removing `rangesToRemove` in `rebalance()` can be optimized

Gas

Issue Description

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2.sol#L270-L313

```
270  function rebalance(
271      Range[] calldata rangesToAdd_,
272      Rebalance calldata rebalanceParams_,
273      Range[] calldata rangesToRemove_
274  ) external onlyManager {
     @@ 275,292 @@
293      for (uint256 i; i < rangesToRemove_.length; i++) {
294          (bool exist, uint256 index) = Position.rangeExist(
295              ranges,
296              rangesToRemove_[i]
297          );
298          require(exist, "RRNE");
299
300          Position.requireNotActiveRange(
301              factory,
302              address(this),
303              address(token0),
304              address(token1),
305              rangesToRemove_[i]
306          );
307
308          for (uint256 j = index; j < ranges.length - 1; j++) {
309              ranges[j] = ranges[j + 1];
310          }
311          ranges.pop();
312      }
313  }
```

The current implementation to remove the `rangesToRemove_` from `ranges` is very

gas-expensive.

And based on the context, it seems unnecessary to maintain the order in `ranges`.

Recommendation

```
293  for (uint256 i; i < rangesToRemove_.length; i++) {
294      (bool exist, uint256 index) = Position.rangeExist(
295          ranges,
296          rangesToRemove_[i]
297      );
298      require(exist, "RRNE");
299
300      Position.requireNotActiveRange(
301          factory,
302          address(this),
303          address(token0),
304          address(token1),
305          rangesToRemove_[i]
306      );
307
308      ranges[index] = ranges[ranges.length - 1];
309      ranges.pop();
310  }
```

Status

✓ Fixed

[WP-G13] `ArrakisV2Resolver.sol#standardBurnParams()` Cache external call results can save gas

Gas

Issue Description

Every call to an external contract costs a decent amount of gas. For optimization of gas usage, external call results should be cached if they are being used for more than one time.

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2Resolver.sol#L145-L238

```
145  function standardBurnParams(uint256 amountToBurn_, IArrakisV2 vaultV2_)
146      external
147      view
148      returns (BurnLiquidity[] memory burns)
149  {
150      uint256 totalSupply = vaultV2_.totalSupply();
151      require(totalSupply > 0, "total supply");
152
153      Range[] memory ranges = vaultV2_.getRanges();
154
155      {
156          UnderlyingOutput memory underlying;
157          (
158              underlying.amount0,
159              underlying.amount1,
160              underlying.fee0,
161              underlying.fee1
162          ) = UnderlyingHelper.totalUnderlyingWithFees(
163              UnderlyingPayload({
164                  ranges: ranges,
165                  factory: factory,
166                  token0: address(vaultV2_.token0()),
167                  token1: address(vaultV2_.token1()),
168                  self: address(vaultV2_)
169              })
170          );
171          underlying.leftOver0 =
172              vaultV2_.token0().balanceOf(address(vaultV2_)) -
```

```
173             vaultV2_.managerBalance0();
174         underlying.leftOver1 =
175             vaultV2_.token1().balanceOf(address(vaultV2_)) -
176             vaultV2_.managerBalance1();
177
178         {
179             uint256 amount0 = FullMath.mulDiv(
180                 underlying.amount0,
181                 amountToBurn_,
182                 totalSupply
183             );
184             uint256 amount1 = FullMath.mulDiv(
185                 underlying.amount1,
186                 amountToBurn_,
187                 totalSupply
188             );
189
190             if (
191                 amount0 <= underlying.leftOver0 &&
192                 amount1 <= underlying.leftOver1
193             ) return burns;
194         }
195     }
196     // #endregion get amount to burn.
197
198     uint128[] memory liquidities = new uint128[](ranges.length);
199     uint256 len;
200     for (uint256 i; i < ranges.length; i++) {
201         uint128 liquidity;
202         {
203             (liquidity, , , , ) = IUniswapV3Pool(
204                 factory.getPool(
205                     address(vaultV2_.token0()),
206                     address(vaultV2_.token1()),
207                     ranges[i].feeTier
208                 )
209             ).positions(
210                     PositionHelper.getPositionId(
211                         address(vaultV2_),
212                         ranges[i].lowerTick,
213                         ranges[i].upperTick
214                     )
215                 );
```


```
216             }
217             liquidities[i] = liquidity;
218
219             if (liquidity != 0) ++len;
220         }
221         burns = new BurnLiquidity[](len);
222         uint256 idx;
223         for (uint256 j; j < ranges.length; j++) {
224             if (liquidities[j] > 0) {
225                 burns[idx] = BurnLiquidity({
226                     liquidity: SafeCast.toUint128(
227                         FullMath.mulDiv(
228                             liquidities[j],
229                             amountToBurn_,
230                             totalSupply
231                         )
232                     ),
233                     range: ranges[j]
234                 });
235                 ++idx;
236             }
237         }
238     }
```

`vaultV2_.token0()` , `vaultV2_.token1()` can be cached in storage to save the external call.

While the `ArrakisV2Resolver.sol#standardBurnParams()` is an external view function, there are on-chain invocations as well, eg:

https://github.com/ArrakisFinance/vault-v2-agreement/blob/46546698096f0f4cd86381cd06be28d43a50a2ea/contracts/functions/FPALMTerms.sol#L27-L30

```
27  BurnLiquidity[] memory burnPayload = resolver.standardBurnParams(
28      balance,
29      vault_
30  );
```

Status

ⓘ Acknowledged

[WP-G14] Avoiding unnecessary storage read can save gas

Gas

Issue Description

https://github.com/ArrakisFinance/v2-palm//blob/0c4572718328e9aa0061c959f658e4d1c4ba4568/contracts/abstracts/PALMTermsStorage.sol#L241-L277

```
241  function setVaultData(address vault_, bytes calldata data_)
242      external
243      override
244      requireAddressNotZero(vault_)
245      requireDelegateWhenDelegateExistsOtherwiseRequireIsOwner(vault_)
246  {
247      IPALMManager(manager).setVaultData(vault_, data_);
248
249      emit LogSetVaultData(
250          delegateByVaults[vault_] != address(0)
251              ? delegateByVaults[vault_]
252              : msg.sender,
253          vault_,
254          data_
255      );
256  }
257
258  /// @notice set Arrakis V2 vault strategy
259  /// @param vault_ Arrakis V2 vault
260  /// @param strat_ strategy to apply during market making
261  /// @dev only callable by delegate of the vault by default or otherwise owner
262  function setVaultStratByName(address vault_, string calldata strat_)
263      external
264      override
265      requireAddressNotZero(vault_)
266      requireDelegateWhenDelegateExistsOtherwiseRequireIsOwner(vault_)
267  {
268      IPALMManager(manager).setVaultStratByName(vault_, strat_);
269
270      emit LogSetVaultStratByName(
271          delegateByVaults[vault_] != address(0)
272              ? delegateByVaults[vault_]
```

```
273                 : msg.sender,
274             vault_,
275             strat_
276         );
277     }
```

Recommendation

```
241     function setVaultData(address vault_, bytes calldata data_)
242         external
243         override
244         requireAddressNotZero(vault_)
245         requireDelegateWhenDelegateExistsOtherwiseRequireIsOwner(vault_)
246     {
247         IPALMManager(manager).setVaultData(vault_, data_);
248         address delegate = delegateByVaults[vault_];
249         emit LogSetVaultData(
250             delegate != address(0)
251                 ? delegate
252                 : msg.sender,
253             vault_,
254             data_
255         );
256     }
```

Status

✓ Fixed

[WP-G15] Storage variables can be cached in memory to save gas

Gas

Issue Description

https://github.com/ArrakisFinance/vault-v2-core/blob/026d9f346394b02b691be2b9259509abe386eab9/contracts/ArrakisV2.sol#L481-L484

```
481     function _applyFees(uint256 fee0_, uint256 fee1_) internal {
482         managerBalance0 += (fee0_ * managerFeeBPS) / hundredPercent;
483         managerBalance1 += (fee1_ * managerFeeBPS) / hundredPercent;
484     }
```

`managerFeeBPS` is a storage variable that is being read multiple times, they can be cached to save ~100 gas for each extra `SLOAD` s.

Recommendation

Change to:

```
481     function _applyFees(uint256 fee0_, uint256 fee1_) internal {
482         uint16 mManagerFeeBPS = managerFeeBPS;
483         managerBalance0 += (fee0_ * mManagerFeeBPS) / hundredPercent;
484         managerBalance1 += (fee1_ * mManagerFeeBPS) / hundredPercent;
485     }
```

Status

✓ Fixed

Appendix

Timeliness of content

The content contained in the report is current as of the date appearing on the report and is subject to change without notice, unless indicated otherwise by WatchPug; however, WatchPug does not guarantee or warrant the accuracy, timeliness, or completeness of any report you access using the internet or other means, and assumes no obligation to update any information following publication.

Disclaimer

This report is based on the scope of materials and documentation provided for a limited review at the time provided. Results may not be complete nor inclusive of all vulnerabilities. The review and this report are provided on an as-is, where-is, and as-available basis. You agree that your access and/or use, including but not limited to any associated services, products, protocols, platforms, content, and materials, will be at your sole risk. Smart Contract technology remains under development and is subject to unknown risks and flaws. The review does not extend to the compiler layer, or any other areas beyond the programming language, or other programming aspects that could present security risks. A report does not indicate the endorsement of any particular project or team, nor guarantee its security. No third party should rely on the reports in any way, including for the purpose of making any decisions to buy or sell a product, service or any other asset. To the fullest extent permitted by law, we disclaim all warranties, expressed or implied, in connection with this report, its content, and the related services and products and your use thereof, including, without limitation, the implied warranties of merchantability, fitness for a particular purpose, and non-infringement. We do not warrant, endorse, guarantee, or assume responsibility for any product or service advertised or offered by a third party through the product, any open source or third-party software, code, libraries, materials, or information linked to, called by, referenced by or accessible through the report, its content, and the related services and products, any hyperlinked websites, any websites or mobile applications appearing on any advertising, and we will not be a party to or in any way be responsible for monitoring any transaction between you and any third-party providers of products or services. As with the purchase or use of a product or service through any medium or in any environment, you should use your best judgment and exercise caution where appropriate. FOR AVOIDANCE OF DOUBT, THE REPORT, ITS CONTENT, ACCESS, AND/OR USAGE THEREOF, INCLUDING ANY ASSOCIATED SERVICES OR MATERIALS, SHALL NOT BE CONSIDERED OR RELIED UPON AS ANY FORM OF FINANCIAL, INVESTMENT, TAX, LEGAL, REGULATORY, OR OTHER ADVICE.